
Why to learn Fortran?

Well suited for numerical computations

– Likely over 50% of scientific applications are written in 
Fortran

Fast code (compilers can optimize well)

Handy array data types

Clarity of code

Portability of code

Optimized numerical libraries available

Short history of Fortran

John W. Backus et al (1954): The IBM Mathematical 
Formula Translating System

Early years development: Fortran II (1958), Fortran IV 
(1961), Fortran 66 & Basic Fortran (1966)

Fortran 77 (1978)

Fortran 90 (1991) a major revision and Fortran 95 (1997) 
a minor revision to it

Short history of Fortran cont.

Fortran 2003 (2004): a major revision, adding e.g. object-
oriented features, C-bindings
– ”Fortran 95/2003” is the current de facto standard

The latest standard is Fortran 2008 (2010): a more minor 
upgrade compared to F03

– Most notable addition: Fortran coarray (CAF) syntax

– Compiler support nearly complete

 Cray: feature complete

 Intel and GNU: most important features (except CAF)

Look and feel

PROGRAM square_root_example

! Comments start with an exclamation point. 
! You will find data type declarations, couple arithmetic operations
! and an interface that will ask a value for these computations.
IMPLICIT NONE
REAL :: x, y   
INTRINSIC SQRT ! Fortran standard provides many commonly used functions

! Command line interface. Ask a number and read it in
WRITE (*,*) 'Give a value (number) for x:'
READ (*,*) x

y = x**2+1   ! Power function and addition arithmetic

WRITE (*,*) 'given value for x:', x
WRITE (*,*) 'computed value of x**2 + 1:', y
! Print the square root of the argument y to screen
WRITE (*,*) 'computed value of SQRT(x**2 + 1):', SQRT(y)

END PROGRAM square_root_example

compiler

linker

source code
(.f, .F, .f90, .F90)

modules
object code

(.o, .so)

libraries
(.a, .so)

executable

INCLUDE
files compiler output

(optional)

linker output
(optional)

Compiling and linking

IMPLICIT NONE

INTEGER :: n0

REAL :: a, b
REAL :: r1=0.0

COMPLEX :: c
COMPLEX :: imag_number=(0.1, 1.0)

CHARACTER(LEN=80) :: place
CHARACTER(LEN=80) :: name='James Bond'

LOGICAL :: test0 = .TRUE.
LOGICAL :: test1 = .FALSE.

REAL, PARAMETER :: pi=3.14159

Variables

Constants defined with the 
PARAMETER clause – they cannot be 
altered after their declaration

Variables must be declared at the 
beginning of the program or 
procedure

They can also be given a value at 
declaration

The intrinsic data types in Fortran are 
INTEGER, REAL, COMPLEX, 
CHARACTER and LOGICAL

PROGRAM numbers
IMPLICIT NONE
INTEGER :: i
REAL :: r
COMPLEX :: c, cc
i = 7
r = 1.618034
c = 2.7182818   !same as c = CMPLX(2.7182818)
cc = r*(1,1) 
CMPLX(r)
WRITE (*,*) i, r, c, cc

END PROGRAM 

Output (one integer, real and two complex values) :
7  1.618034  (2.718282, 0.000000)  (1.618034, 1.618034)

Assignment statements

How can I convert numbers to 
character strings and vice versa? See 
“INTERNAL I/O” in the File I/O lecture.

Automatic change of representation, 
works between all numeric intrinsic 
data types

Arithmetic operators
REAL :: x,y
INTEGER :: i = 10
x = 2.0**(-i)  !power function and negation     precedence: first
x = x*REAL(i)  !multiplication and type change  precedence: second
x = x/2.0      !division                        precedence: second
i = i+1        !addition                        precedence: third
i = i-1        !subtraction                     precedence: third

Relational operators
<  or .LT.   !less than
<= or .LE.   !less than or equal to
== or .EQ.   !equal to
/= or .NE.   !not equal to
>  or .GT.   !greater than
>= or .GE.   !greater than or equal to

Logical operators
.NOT.      !logical negation              precedence: first
.AND.      !logical conjunction           precedence: second
.OR.       !logical inclusive disjunction precedence: third

Operators

Arrays

INTEGER, PARAMETER :: M = 100, N = 500
INTEGER :: idx(M)
REAL :: vector(0:N-1)
REAL :: matrix(M, N)
CHARACTER (len = 80) :: screen ( 24)

! or

INTEGER, DIMENSION(1:M) :: idx
REAL, DIMENSION(0:N-1) :: vector
REAL, DIMENSION(M, N) :: matrix
CHARACTER(len=80), dimension(24) :: screen

By default, indexing starts from 1

Control structures: IF conditionals

PROGRAM test_if
IMPLICIT NONE
REAL :: x, y, eps, t

WRITE(*,*)' Give x and y :'
READ(*,*) x, y
eps = EPSILON(x)

IF (ABS(x) > eps) THEN
t = y/x

ELSE
WRITE(*,*)'division by zero'
t = 0.0

END IF
WRITE(*,*)' y/x = ',t

END PROGRAM

Control structures: DO loops

! DO loop with an integer counter (count controlled)

INTEGER :: i, stepsize, NumberOfPoints
INTEGER, PARAMETER :: max_points=100000
REAL :: x_coodinate(max_points), x, totalsum

stepsize = 2

DO i = 1, NumberOfPoints, stepsize
x_coordinate(i) = i*stepsize*0.05

END DO

! Condition controlled loop (DO WHILE)
totalsum = 0.0

READ(*,*) x

DO WHILE (x > 0)
totalsum = totalsum + x
READ(*,*) x

END DO

Control structures: DO loops

! DO loop without loop control

REAL :: x, totalsum, eps

totalsum = 0.0

DO
READ(*,*) x
IF (x < 0) THEN

EXIT ! exit the loop
ELSE IF (x > upperlimit) THEN

CYCLE ! do not execute any statements but
! cycle back to the beginning of the loop

END IF
totalsum = totalsum + x

END DO

SELECT CASE statements 
matches the entries of a 
list against the case index

– Only one found match is 
allowed

– Usually arguments are 
character strings or 
integers

– DEFAULT branch if no 
match found

... 
INTEGER :: i
LOGICAL :: is_prime, 

test_prime_number
...
SELECT CASE (i)

CASE (2,3,5,7)
is_prime = .TRUE.

CASE (1,4,6,8:10)
is_prime = .FALSE.

CASE DEFAULT
is_prime=test_prime_number(i)

END SELECT
...

Control structures: SELECT CASE

Control structures: example 1

PROGRAM gcd
! Computes the greatest common divisor, Euclidean algorithm
IMPLICIT NONE
INTEGER :: m, n, t
WRITE(*,*) 'Give positive integers m and n :'
READ(*,*) m, n
WRITE(*,*)'m:', m,' n:', n
positive_check: IF (m > 0 .AND. n > 0) THEN

main_algorithm: DO WHILE (n /= 0)
t = MOD(m,n)
m = n
n = t

END DO main_algorithm
WRITE(*,*) 'Greatest common divisor: ',m

ELSE
WRITE(*,*) 'Negative value entered'

END IF positive_check

END PROGRAM gcd

Labels can be given to 
control structures and used 
in conjunction with e.g. exit
and cycle statements

Control structures: 
example 2

PROGRAM placetest
IMPLICIT NONE
LOGICAL :: in_square1, in_square2
REAL :: x,y
WRITE(*,*) 'Give point coordinates x and y'
READ (*,*) x, y
in_square1 = (x >= 0. .AND. x <= 2. .AND. y >= 0. .AND. y <= 2.)
in_square2 = (x >= 1. .AND. x <= 3. .AND. y >= 1. .AND. y <= 3.)
IF (in_square1 .AND. in_square2) THEN      ! inside both 

WRITE(*,*) 'Point within both squares'
ELSE IF (in_square1) THEN                  ! inside square 1 only

WRITE(*,*) 'Point inside square 1'
ELSE IF (in_square2) THEN                  ! inside square 2 only

WRITE(*,*) 'Point inside square 2'
ELSE                                       ! both are .FALSE.

WRITE(*,*) 'Point outside both squares'
END IF

END PROGRAM placetest 

Source code remarks

A variable name can be no longer than 31 characters 
(containing only letters, digits or underscore, must start with a 
letter)

Maximum row length is 132 characters 

There can be max 39 continuation lines
– if a line is ended with ampersand (&), the line continues onto the next 

line. 

No distinction between lower and uppercase characters
– Character strings are case sensitive 

Semicolon (;) as a separator between statements on a single 
line

Source code remarks: example 1

! Character strings are case sensitive
CHARACTER(LEN=32) :: ch1, ch2
Logical :: ans
ch1 = 'a'
ch2 = 'A'
ans = ch1 .EQ. ch2
WRITE(*,*) ans     ! OUTPUT from that WRITE statement is: F

! When strings are compared 
! the shorter string is extended with blanks 
WRITE(*,*) 'A' .EQ. 'A '    ! OUTPUT: T
WRITE(*,*) 'A' .EQ. ' A'    ! OUTPUT: F

Source code remarks: example 2

INTEGER :: a, b, c, d

! Statement separation: newline and semicolon, ;
! Semicolon as a statement separator
a = a * b; c = d**a
! The above is equivalent to following two lines
a = a * b
c = d**a

Summary

Fortran 95/2003 – despite its long history – is a modern 
programming language targeting especially scientific 
computing

– Versatile, quite easy to learn, powerful

In our first encounter, we discussed

– Variables & data types

– Control structures: loops and conditionals

– Operators

Fortran arrays

22

Outline

Introduction to Fortran arrays

Array declaration and syntax

Array initialization

Array sections

Introduction to Fortran arrays

Arrays enable a natural way to access vector and/or 
matrix data during computation

Fortran language is a very versatile in handling especially 
multi-dimensional arrays (unlike C or some other 
languages)

Consider the following code snippet for Matrix-Vector 
multiply as an example

Introduction to Fortran arrays

Arrays are declared in a pretty much similar fashion to 
scalar variables

They all refer to a particular built-in data type (or derived 
data type) , but …

… they all have one or more dimensions specified in the 
variable declaration

Fortran supports up to 15 dimensions

– Note: This includes F2008 Coarray (CAF) notation, too

In Fortran 77 and before, 
arrays were traditionally 
accessed element-by-
element – basis

INTEGER, PARAMETER :: M = 3,  N = 4
REAL(kind=8) :: A(M,N), x(N), y(M)
INTEGER :: I , J

DO I=1,M; y(I) = 0; END DO

OUTER_LOOP: DO J = 1, N
INNER_LOOP : DO  I = 1, M
y(I)= y(I) + A(I,J)*x(J)

END DO INNER_LOOP
END DO OUTER_LOOP

Arrays in old Fortran

Arrays in modern Fortran 
enable a more natural way 
to express mathematical 
operations

𝑦 = 𝐴𝑥 = 

𝑗=1

𝑁

𝑎𝑗𝑥𝑗⟺

𝑦1
𝑦2
𝑦3
=

𝑎11 𝑎12 𝑎13 𝑎14
𝑎21 𝑎22 𝑎23 𝑎24
𝑎31 𝑎32 𝑎33 𝑎34

𝑥1
𝑥2
𝑥3
𝑥4

INTEGER, PARAMETER :: M = 3,  N = 4
REAL(kind=8) :: A(M,N), x(N), y(M)
INTEGER :: J

y(:)=0
LOOP: do J = 1,N

y(:) = y(:) + A(:,J)*x(J)
end do LOOP

Arrays in modern Fortran

Array syntax

In older Fortran, arrays were traditionally accessed 
element-by-element basis

Modern Fortran has a way of accessing several elements 
in one go: array syntax

y(:) = y(:) + A(:,j)*x(j)

Array syntax improves code readability and sometimes 
even performance through better code optimization

28

Array declaration: example

INTEGER, PARAMETER :: M = 100, N = 500
INTEGER :: idx(M)
REAL(kind=4) :: vector(0:N-1)
REAL(kind=8) :: matrix(M,N)
CHARACTER(len=80) :: screen(24)
TYPE(my_own_type) :: object(10)

! Or equivalently

INTEGER, DIMENSION(M) :: idx
REAL(kind=4), DIMENSION(0:N-1) :: vector
REAL(kind=8), DIMENSION(1:M,N) :: matrix
CHARACTER(len=80), DIMENSION(24) :: screen
TYPE(my_own_type), DIMENSION(1:10) :: object

Data layout in multi-dimensional arrays

Always increment the left-most index of multi-
dimensional arrays in the innermost loop (i.e. fastest)

– “Column major” ordering in Fortran vs. “Row major” 
ordering in C

Some compilers (with sufficient optimization flags) may 
re-order loops automatically

do i=1,N 
do j=1,M
y(i) = y(i)+ a(i,j)*x(j)

end do
end do

do j=1,M 
do i=1,N
y(i) = y(i)+ a(i,j)*x(j)

end do
end do

Array syntax: summary

! Explicit DO loops

integer :: m = 100, n = 200

real    :: a(m,n), x(n), y(m)

integer :: i, j

y = 0.0

outer_loop: do j = 1, n
inner_loop: do i = 1, m

y(i) = y(i) + a(i,j) * x(j)
end do inner_loop

end do outer_loop

! Array syntax

integer :: m = 100, n = 200

real    :: a(m,n), x(n), y(m)

integer :: j

y = 0.0

outer_loop: do j = 1, n
y(:) = y(:) + a(:,j) * x(j)

end do outer_loop

31

Array initialization

To make a program meaningful, we need to feed its 
variables with some values

Arrays can be initialized element-by-element basis, 
copied from another array, or by using single line data 
initialization statements

More advanced initialization involves use of FORALL and 
WHERE –statements, or use of RESHAPE –intrinsic 
function

Array initialization

! Element-by-element initialization

do J=1, N
idx(J)=J
vector(J)=0

enddo

! Initialization by copying from another array

REAL(kind=8) :: to(100,100),  from(0:199,0:199)

to(1:100, 1:100) = from(0:199:2,0:199:2)

Every 2nd element

Array initialization

INTEGER, PARAMETER :: fixed(2:4) = (/ 20, 30, 40 /)
INTEGER :: values(1:3) = [11, -22, 33] 

INTEGER :: idx(0:10)
! Initialization of idx
DATA  idx / 0, 1, 2, 3, 7 * 0 /
! or preferably
idx(0:10) = (/ 0, (j, i = 1, 3), (0, j = 4,10) /)
! or equivalently (F03)
idx(0:10) = [0, (j, i = 1, 3), (0, j = 4,10)]

Fortran array syntax 
enables accessing a part 
of an array in an intuitive 
way

Array sections make 
Fortran very suitable for 
scientific computing

! Set elements from 3 to N+8 to 0
Sub_Vector(3:N+8)=0

! Set every third element from 1 to
! 3*N+1 to 1
Every_Third(1:3*N+1:3) = 1

! Set block [i-1:i+1,j-2:j+2] to k
Diag_Block(i–1:i+1,j–2:j+2) = k

Arrays sections

For instance, sections 
enable referring to a sub-
block of a matrix or a sub-
cube of a 3D array

REAL(kind=8) :: A(1000, 1000)
INTEGER(kind=2) :: &

pixel_3D(256, 256, 256)

! Access subblock of a matrix
A(2:500,3:300:3)=4.0_8

! Access subcube of a 3D array
pixel_3D(128:150,56:80,1:256:8)=32000

Array sections

When copying array 
sections, both left -and 
right hand sides of the 
assignment statement 
must have conforming 
dimensions

! Conforming size of 3-by-10
LHS(1:3, 0:9)=RHS(-2:0, 20:29)

! Error: LHS 2-by-10, RHS 3-by-10
LHS(1:2, 0:9) = RHS(-2:0, 20:29)

Array sections

Array sections can be 
passed into a procedure

Be aware that an array 
section is usually copied
into a hidden temporary 
array upon calling a 
procedure and copied 
back to the array section 
upon return

INTEGER :: Array(10, 20)

! Pass a full array
CALL SUB(Array)

! Pass a subblock
CALL SUB(Array(5:10,10:20))

! Pass a non-contiguous subblock
CALL SUB(Array(1:10:2,1:1))

! Pass an array slice
CALL SUB(Array(1:4,1:))
CALL SUB(Array(:10,:)) 

Arrays sections

Summary

Use of arrays makes Fortran language a very versatile 
vehicle for computationally intensive program 
development

Using array syntax, vectors and matrices can be initialized 
and used in a very intuitive way – and thus suitability for 
scientific computing applications is well justified

Use of array sections increase code readability and 
usually reduce chances of mistakes

Procedures and modules

40

Outline

Structured programming

Procedures: functions and subroutines

Interfaces

Procedure arguments

Modules

Structured programming

Structured programming based on program sub-units 
(functions, subroutines and modules) enables

– testing and debugging separately

– re-use of code

– improved readability

– re-occurring tasks

The key to success is in well defined data structures and 
scoping, which lead to clean procedure interfaces

What are procedures?

With procedures we mean subroutines and functions

Subroutines exchange data through its argument lists
CALL mySubroutine(arg1, arg2, arg3)

Functions return a value

value = myFunction(arg1, arg2)

Both can also interact with the rest of the program 
through module (global) variables

Procedure declarations

Function

Declaration:

[TYPE] FUNCTION func(arg1,
arg2,...) [RESULT(val)]

[declarations]
[statements]

END FUNCTION func

Call convention:
res = func(arg1, arg2,...)

Subroutine

Declaration:

SUBROUTINE sub(arg1, arg2,...)

[declarations]
[statements]

END SUBROUTINE sub

Call convention:
CALL sub(arg1, arg2,...)

Procedure declarations: example

REAL FUNCTION dist(x, y)
IMPLICIT NONE
REAL :: x, y
dist = SQRT(x**2 + y**2)
END FUNCTION dist

PROGRAM do_something
...
r = dist(x, y)
...

SUBROUTINE dist(x, y, d)
IMPLICIT NONE
REAL :: x, y, d
d = SQRT(x**2 + y**2)
END SUBROUTINE dist

PROGRAM do_something
...
call dist(x, y, r)
...

Procedure types

There are four procedure types in Fortran 90: intrinsic, 
external, internal and module procedures

Procedure types differ in

– Scoping, i.e. what data and other procedures a procedure 
can access

– Interface type, explicit or implicit

In Fortran the procedure arguments are always passed by 
reference, i.e. just as a pointer to a memory location

Compiler can check the argument types of the at compile 
time only if the interface is explicit

Procedure types

The interfaces of the intrinsic, internal and module 
procedures are explicit

The interfaces of the external procedures, such as many 
library subroutines, are implicit. You can write an explicit 
interface to those, though.

Intrinsic procedures are the procedures defined by the 
programming language itself, such as INTRINSIC SIN

Internal procedures

Each program unit (program/subroutine/function) may 
contain internal procedures

SUBROUTINE mySubroutine

...
CALL myInternalSubroutine
...

CONTAINS
SUBROUTINE myInternalSubroutine
...
END SUBROUTINE myInternalSubroutine

END SUBROUTINE mySubroutine

Internal procedures

Declared at the end of a program unit after the 
CONTAINS statement

– Nested CONTAINS statements are not allowed

Variable scoping:

– Parent unit’s variables and objects are accessible

– Parent unit’s variables are overlapped by local variables 
with the same name

Often used for ”small and local, convenience” 
subroutines within a program unit

Internal procedures: example

SUBROUTINE parent()
IMPLICIT NONE
INTEGER :: i,j

i = 1; j = 1
CALL child()
! After subroutine call i = 2 and j = 1
CONTAINS

SUBROUTINE child()
IMPLICIT NONE
integer :: j
i = i + 1 ! Variable i is from the scope of parent
j = 0  ! Variable j has local scope

END SUBROUTINE child

END SUBROUTINE parent

External procedures

Declared in a separate program unit

– Referred to with the EXTERNAL keyword 

– Compiled separately and linked to the final executable

Avoid using them within a program, module procedures 
provide much better compile time error checking

External procedures are often needed when using

– procedures written with different programming language

– library routines (e.g. BLAS and MPI libraries)

– old F77 subroutines

Wrong calling arguments 
to EXTERNAL procedures 
may lead to errors during 
the executable linking 
phase or even when the 
executable is being run

It is highly recommended 
to construct INTERFACE
blocks for any external 
procedures used

interface
subroutine not_dangerous(a, b, c)
integer :: a, b, c
end subroutine not_dangerous

end interface

integer :: x, y, z

x=1; y=1; z=1

! Call external subroutine without
! an interface

call dangerous(x,y,z)

! Call external subroutine with
! an interface

call not_dangerous(x,y,z)

Interfaces

Interfaces

For external procedures, interfaces determine the type 
and properties of arguments and return values

Defined by an INTERFACE block:
interface 
interface-body
end interface

The interface-body matches the subprogram header

– position, rank and type of arguments

– return value type and rank (for functions)

Interfaces: example

! LU decomposition from LAPACK

interface
subroutine DGETRF(M, N, A, LDA, IPIV, INFO)
integer :: INFO, LDA, M, N
integer :: IPIV(*)
double precision :: A(LDA,*)

end subroutine DGETRF

end interface

! Euclidean norm from BLAS

interface
function DNRM2( N, X, INCX )
integer :: N, INCX
double precision :: X(*)
double precision :: DNRM2

end function DNRM2

end interface

54

Procedure arguments

Fortran passes call arguments by reference:

– Means that only the memory addresses of the arguments 
are passed to the called procedure

– Any change to argument changes the actual argument

– Compiler can check the argument types only if the 
interface is explict, i.e. compiler has information about the 
called procedure at compile time.

– The INTENT keyword increases readability and enables 
better compile-time error checking

Declares how formal argument 
is intended for transferring a 
value

– IN: the value of the 
argument read/only ie. 
cannot be changed

– OUT: the value of the 
argument must be provided

– INOUT (the default)

Compiler uses INTENT for error 
checking and optimization

SUBROUTINE foo(x, y, z)
IMPLICIT NONE
REAL, INTENT(IN) :: x
REAL, INTENT(INOUT) :: y
REAL, INTENT(OUT)   :: z

x = 10 ! Compilation error
y = 10 ! Correct
z = y * x ! Correct

END SUBROUTINE foo

INTENT keyword

Passing array arguments

Two (three) ways to pass arrays to procedures

– Explicit shape array (dimensions passed explicitly, F77-style)
subroutine foo(size1, size2, ..., matrix, ...)

implicit none
integer :: size1, size2
real, dimension(size1, size2) :: matrix
...

– Assumed shape array (requires explicit interface)
subroutine foo(matrix)

real, dimension(:,:) :: matrix

 One can use the intrinsic function SIZE for checking the 
actual dimensions

We may pass into 
procedures also other 
procedures (i.e., not only 
data)

Internal procedures 
cannot be used as 
arguments in F95/F03 
(restriction lifted in F08)

PROGRAM degtest
IMPLICIT NONE
INTRINSIC ASIN, ACOS
WRITE (*,*) 'arcsin(0.5): ', & 

deg(ASIN,0.5)
WRITE (*,*) 'arccos(0.5): ', & 

deg(ACOS,0.5)

CONTAINS
REAL FUNCTION deg(f, x)
IMPLICIT NONE
INTRINSIC ATAN 
REAL, EXTERNAL :: f
REAL, INTENT(IN) :: x
deg = 45 * f(x) / ATAN(1.0)

END FUNCTION deg

END PROGRAM degtest

Procedure arguments

Modular programming

Modularity means dividing a program into minimally 
dependent modules

– Enables division of the program into smaller self-contained 
units

Where to employ Fortran modules

– Global definitions of procedures, variables and constants

– Compilation-time error checking

– Hiding implementation details

– Grouping routines and data structures

– Defining generic procedures and custom operators

Module procedures and variables

Declaration

MODULE check
IMPLICIT NONE
INTEGER, PARAMETER :: &
longint = SELECTED_INT_KIND(8)

CONTAINS
FUNCTION check_this(x) RESULT(z)
INTEGER(longint):: x, z
...

END FUNCTION
END MODULE check

Usage

PROGRAM testprog
USE check
IMPLICIT NONE
INTEGER(KIND=longint) :: x,test
test=check_this(x)

END PROGRAM testprog

A good habit
USE check, ONLY: longint

Procedures defined in 
modules can be referred 
to in any other program 
unit with the USE clause

Module procedures are 
declared after the CONTAINS 
statement

Global data and global variables

Global variables can be accessed from any program unit

F90 module variables provide controllable way to define 
and use global variables

MODULE commons
INTEGER, PARAMETER :: r = 0.42
INTEGER, SAVE :: n, ntot
REAL, SAVE :: abstol, reltol

END MODULE commons

– Explicit interface: type checking, limited scope

Implemented as common blocks in old F77 codes
COMMON/EQ/N,NTOT
COMMON/TOL/ABSTOL,RELTOL

– Extremely error prone
61

Visibility of module objects

Variables and procedures in modules can be PRIVATE or 
PUBLIC

– PUBLIC = visible for all program units using the module 
(the default)

– PRIVATE will hide the objects from other program units

REAL :: x, y
PRIVATE :: x
PUBLIC :: y

Program units

Module procedures Internal procedures

ModulesMain program

External 
procedures

Summary

Procedural programming makes the code more readable 
and easier to develop

– Procedures encapsulate some piece of work that makes 
sense and may be worth re-using elsewhere

Fortran uses functions and subroutines

– Values of procedure arguments may be changed upon 
calling the procedure

Fortran modules are used for modular programming and 
data encapsulation

More about Fortran arrays

Array intrinsic functions

Dynamic memory allocation

Fortran POINTER -attribute

Array intrinsic functions

Array intrinsic functions are built-in functions which can 
apply various operations on the whole array at once, i.e., 
not just individual array elements

As a result another array or just a scalar value is returned

A subset selection through masking is also possible

– Masking and use of array (intrinsic) functions is often 
accompanied with use of FORALL and WHERE array 
statements

– Operations are performed for those elements where 
corresponding elements of the mask are .TRUE.

Array intrinsic functions

The most commonly used array intrinsic functions are:

– SIZE, SHAPE, COUNT, SUM

– ANY, ALL

– MINVAL, MAXVAL , MINLOC, MAXLOC

– RESHAPE

– DOT_PRODUCT, MATMUL, TRANSPOSE

– PACK, UNPACK, SPREAD

Array intrinsic functions: 
SIZE, SHAPE, COUNT, SUM

SIZE(array [, dim]) returns # of elements in the 
array [, along the specified dimension]

SHAPE(array) returns an INTEGER vector containing 
the size of array in each dimension

COUNT(L_array [,dim]) returns count of elements 
which are .TRUE. in L_array

SUM(array[, dim][, mask]) sum of the elements of 
array [, along dimension] [, under mask]

Array intrinsic functions: ANY, ALL

ANY(L_array [, dim]) returns a scalar value of 
.TRUE. if any value in L_array is .TRUE.

ALL(L_array [, dim]) returns a scalar value of 
.TRUE. if all values in L_array are .TRUE.

Array intrinsic functions:
MINVAL, MAXVAL, MINLOC, MAXLOC

MINVAL(array [,dim] [, mask]) returns the 
minimum value of a given array 
[, along the specified dimension] and [, under mask]

MAXVAL is the same as MINVAL, but returns the 
maximum value of a given array

MINLOC(array [, mask]) returns a vector of 
location(s) [, under mask], where the minimum value(s) 
is/are found

MAXLOC similar to MINLOC, but for maximums

Array intrinsic functions: example

INTEGER :: J
INTEGER, PARAMETER :: M = 10, N = 20
REAL :: X(M,N), V(N)
CALL random_number(X(:,:))

PRINT *, SIZE(X), SIZE(V) ! Prints M * N, N
PRINT *, SHAPE(X) ! Prints M, N
PRINT *, SIZE(SHAPE(X))   ! Prints 2
PRINT *, COUNT(X >= 0)
PRINT *, SUM(X, DIM=2, MASK=X < 0.5)  ! The result is a vector !!

V(1:N) = [ (J, J=1,N) ]
PRINT *, ANY(V > -1 .and. V < 1) 
PRINT *, ALL(X >= 0, DIM=1)
PRINT *, MINVAL(V), MAXVAL(V)
PRINT *, MINLOC(V), MAXLOC(V)

Array intrinsic functions: RESHAPE

RESHAPE(array, shape) returns a reconstructed (=a 
copy of an) array with different shape than in the input 
array

– Can be used as a single line statement to initialize an array 
(sometimes at the expense of readability)

Array intrinsic functions: RESHAPE example

INTEGER :: j, IA(4, 2)

! Initialize each column separately
IA(:,1)=[ (j, j = 1,SIZE(IA,dim=1)) ]
IA(:,2)=[ (SIZE(IA,dim=1)+j, j = 1, SIZE(IA,dim=1)) ]

! Equivalently with RESHAPE
IA=RESHAPE( [ (j, j = 1, SIZE(IA)) ], SHAPE(IA) )

IA=

𝟏 𝟓
𝟐 𝟔
𝟑 𝟕
𝟒 𝟖

Array intrinsic functions:
DOT_PRODUCT, MATMUL, TRANSPOSE, 

DOT_PRODUCT(a_vec, b_vec) returns a scalar dot 
product of two vectors

MATMUL(a_mat, b_mat) returns a matrix containing 
matrix multiply of two matrices

TRANSPOSE(a_mat) returns a transposed matrix of the 
input matrix

75

Array intrinsic functions: example

INTEGER :: L, M, N
REAL :: A(L,M), B(M,N), C(L,N)
REAL :: A_tr(M,L)
REAL :: V1(N), V2(N), DOTP

! Transpose a matrix
A_tr = TRANSPOSE(A)
! Compute matrix-matrix product C=A*B
C = MATMUL(A, B)
! Compute dot product (v1,v2)=v2^T*v1
DOTP = DOT_PRODUCT(V1, V2)

Array intrinsic functions:
PACK, UNPACK, SPREAD

PACK(a, mask[, vector]) includes elements of a
under mask and optionally places them to vector. The 
reverse is called UNPACK

SPREAD(a, dim, ntimes) replicates array a along the 
dim times ntimes

Array intrinsic functions: example

INTEGER :: V(2) = [1, 2]
LOGICAL :: LMASK(SIZE(V)*SIZE(V))
INTEGER, DIMENSION(SIZE(V),SIZE(V)) :: DIAGMAT, FIELD

! Initialize mask
LMASK(1:SIZE(V)*SIZE(V)) = [ .TRUE., .FALSE., .FALSE., .TRUE. ]
FIELD = 0

! Construct a 2-by-2 diagonal matrix
DIAGMAT = UNPACK(V, RESHAPE(LMASK, SHAPE(FIELD)), FIELD)
! Construct three copies of the rows of V
PRINT *, SPREAD(V, 1, 3)  ! Prints ”1 1 1 2 2 2”
! Get nonzero values of DIAGMAT
PRINT *, PACK(RESHAPE(DIAGMAT, SHAPE(LMASK)), LMASK) ! Prints “1 2”

Array intrinsic functions

Array control statements FORALL and WHERE are 
commonly used in the context of manipulating arrays

– Technically they are not array intrinsic functions, but very 
closely related to array manipulations

FORALL and WHERE can provide masked assignment of 
values using efficient vector operations

Relicts from HPF – a failed attempt to introduce parallel 
Fortran: ”High Performance Fortran” 

Array intrinsic functions: example

INTEGER :: j, ix(5)

ix(:) = (/ (j, j=1, size(ix)) /)

WHERE (ix == 0) ix = -9999

WHERE (ix < 0) 
ix = -ix

ELSEWHERE
ix = 0

END WHERE

INTEGER :: j

REAL :: a(100,100), b(100), c(100)

! Fill in diagonal matrix

FORALL (j=1:100) a(j,j) = b(j)

! Fill in lower bi-diagonal matrix

FORALL (j=2:100) a(j,j-1) = c(j)

Dynamic memory allocation

Size of an array may be static or dynamic

For small array sizes a static dimensioning is usually not a 
problem

For large arrays dynamic memory allocation is maybe the 
only option – or otherwise the program may not fit into 
the  memory – and you will not be able to run it

Dynamic memory allocation

Fortran provides two different mechanisms to 
dynamically allocate memory for arrays:

1. Array variable declaration has an ALLOCATABLE (or a 
POINTER) attribute, and memory is allocated through 
ALLOCATE –statement 

2. Array variable, which is declared in the procedure with 
(runtime) size information coming from the procedure’s 
argument list or from a Fortran MODULE, is called 
automatic array: no ALLOCATE is needed

Dynamic memory allocation: example

INTEGER :: M, N, alloc_stat
INTEGER, ALLOCATABLE :: idx(:)
REAL(kind = 8), ALLOCATABLE :: mat(:,:)

M = 100 ; N = 200

ALLOCATE( idx(0:M–1) , STAT = alloc_stat )
IF (alloc_stat /= 0) STOP ! An error

ALLOCATE( mat(M, N) , STAT = alloc_stat )
IF (alloc_stat /= 0) STOP ! An error

DEALLOCATE(idx , mat)

When automatic arrays 
are being used, no explicit 
ALLOCATE/DEALLOCATE
is needed

SUBROUTINE SUB (M)
USE some_module, ONLY : N
INTEGER, INTENT(IN) :: M

INTEGER :: idx(0:M–1)
REAL(kind = 8) :: mat(M , N)

! Implementation omitted
END SUBROUTINE SUB

Dynamic memory allocation: example

Dynamic memory allocation

When using ALLOCATE-statement, it is always 
recommended to use ALLOCATABLE rather than 
POINTER -attribute in dynamic variable declaration

To avoid unexpected memory growth (”memory leak”), 
please do remember to use DEALLOCATE for every 
ALLOCATE statement ever used

– This has been relaxed now and some newer Fortran 
compilers will automatically de-allocate ALLOCATABLE
(but not POINTER) arrays, when they go out of scope

85

Pointers to arrays

The POINTER attribute enables to create array (or scalar) 
aliasing variables

Pointer variables are usually employed to refer to 
another array or array section

A pointer variable can also be a sole variable itself, but 
requires ALLOCATE
– Not recommended; the ALLOCATABLE attribute and 

employ POINTER variables for aliasing only in stead

Note for C-programmers: a "pointer" has a slightly 
different meaning in C and Fortran

Pointers to arrays

A POINTER can refer to an already allocated memory 
region

INTEGER, POINTER :: p_x(:) => NULL()
INTEGER, TARGET :: x(1000)
...
p_x => x
p_x => x(2 : 300)
p_x => x(1 : 1000 : 5)
...
p_x(1) = 0
NULLIFY(p_x)

Disconnects p_x from x

Initialized to point to nothing

This would also change x(1) to 0

Pointers provide a neat way for array
sections

ASSOCIATED –function 
can be used to check 
whether a POINTER is 
associated with a target

– Returns .TRUE. if 
associated with a target, 
.FALSE. if not

– For an uninitialized 
POINTER variables, the 
return value is undefined

REAL(kind = 8), POINTER :: &
p_mat(: ,:) => NULL()

REAL(kind = 8), TARGET :: &
mat(100,200)

p_mat => mat
IF ( ASSOCIATED (p_mat) ) & 

PRINT *, 'Points to something'

NULLIFY(p_mat)
IF (.NOT. ASSOCIATED (p_mat) ) &

PRINT *, 'Points to nothing'

Pointers to arrays

Summary

Array intrinsic functions further simplify coding efforts 
and improve program code readability when using 
Fortran arrays

Dynamic memory allocation enables sizing of arrays 
according to particular needs 

Pointers offer a versatile alias mechanism to refer into 
the existing arrays or array sections

Input/Output

Outline

Input/output (I/O) formatting

Internal I/O

File I/O

– File opening and closing

– Writing and reading to/from a file

– Formatted and unformatted (binary) files

– Stream I/O

Input/Output formatting

To prettify output and to make it human readable, use 
FORMAT descriptors in connection with the WRITE
statement

Although less often used nowadays, it can also be used 
with READ to input data at fixed line positions and using 
predefined field lengths

Use through FORMAT statements, CHARACTER variable or 
embedded in READ/WRITE fmt keyword

Data type FORMAT descriptors Examples

Integer Iw, Iw.m WRITE(*,'(I5)') J
WRITE(*,'(I5.3)') J
WRITE(*,'(I0)') J

Real (decimal and 
exponential forms,
auto-scaling)

Fw.d
Ew.d
Gw.d

WRITE(*,'(F7.4)') R
WRITE(*,'(E12.3)') R
WRITE(*,'(G20.13)') R

Character A, Aw WRITE(*,'(A)') C

Logical Lw WRITE(*,'(L2)') L

w=width of the output field, d=number of digits to the right of decimal 
point, m=minimum number of characters to be used. 
Variables: Integer :: J,  Real :: R, Character :: C, Logical :: L

Output formatting

Output formatting: miscellaneous

With complex numbers provide format for both real and 
imaginary parts:

COMPLEX :: Z
WRITE (*,'(F6.3,2X,F6.3)') Z  ! Real & Imaginary parts

Line break, whitespace, tabbing:
WRITE (*,'(F6.3,/,F6.3)') X, Y ! X & Y in separate lines

WRITE (*,'(I3,2X,F6.3)') I, X ! 2 spaces between I & X
WRITE (*,'(I5,T20,I5)') I, J ! 20 tabs between I & J

It is possible that an edit descriptor will be repeated a 
specified number of times

WRITE (*,'(5I8)') IVEC(1:5)
WRITE (*,'(4(I5,2X,F8.3))') (IVEC(J),ZVEC(J),J=1,4)

I0 and G0 edit descriptors in FORMAT

Dynamic sizing of REAL and INTEGER valued output 
– I0 appeared in F03 and G0 was introduced in F08

Correspond to C-language %d and %g fprintf–formats

Output fields are left justified with all the unnecessary leading 
blanks (and precision for REAL valued variables) removed
INTEGER :: I = 12345
REAL (kind=4) :: SP = 1.23e0
REAL (kind=8) :: DP = 1.234567890d0
WRITE(*,fmt='("<I=",I0,", REALs=",2(G0,1X),">")') I,SP,DP

Output is (the “˽” denotes a space character)
<I=12345,˽REALs=1.230000˽1.234567890000000˽>

Internal  I/O

Often it is necessary to filter out data from a given 
character string

Or to pack values into a character string

Fortran internal I/O with READ & WRITE becomes 
now handy

Actual (physical) files are not involved at all

Internal  I/O: examples

CHARACTER(LEN=13) :: CL1
CHARACTER(LEN=60) :: CL2
INTEGER :: njobs, ISTEP

! Extract a number from character string
CL1 = 'Time step# 10'
READ(CL1,fmt='(10X,I3)') ISTEP

! Write data to a character string
njobs = 2014
WRITE(CL2,'(A,I0)') 'The number of jobs completed = ', njobs

Opening and closing files: basic concepts

Writing to or reading from a file is similar to writing onto 
a terminal screen or reading from a keyboard

Differences

– File must be opened with an OPEN statement, in which the 
unit number and (optionally) the file name are given

– Subsequent writes (or reads) must to refer to the given 
unit number 

– File should be closed at the end

Opening and closing a file

The syntax is (the brackets [ ] indicate optional keywords 
or arguments)

OPEN([unit=]iu, file='name' [, options])

CLOSE([unit=]iu [, options])

For example
OPEN(10, file= 'output.dat', status='new')

CLOSE(unit=10, status='keep')

Opening and closing a file

The first parameter is the unit number

The keyword unit= can be omitted

The unit numbers 0, 5 and 6 are predefined

– 0 is output for standard (system) error messages

– 5 is for standard (user) input

– 6 is for standard (user) output

– These units are opened by default and should not be re-
opened nor closed by the user

Opening and closing a file

The default input/output unit can be referred with a star:
WRITE(*, ...)

READ(*, ...)

– Note that these are not necessarily the same as the stdout
and stdin unit numbers 6 and 5

If the file name is omitted in the OPEN, the file name will 
based on unit number being opened, e.g. for unit=12
this usually means the filename ’fort.12’ (on UNIX-
systems)

File opening options

STATUS: existence of a file

– 'old', 'new', 'replace', 'scratch', 'unknown'

POSITION: offset, where to start writing

– 'append'

ACTION: file operation mode

– 'write', 'read', 'readwrite'

FORM: text or binary file 

– 'formatted', 'unformatted'

File opening options

ACCESS: direct or sequential file access

– 'direct', 'sequential', 'stream',

IOSTAT: error indicator, (output) integer

– Non-zero only upon an error

ERR: the Fortran label number to jump upon an error

RECL: record length, (input) integer

– For direct access files only

– Warning (check): may be in bytes or words

File opening: file properties

Use INQUIRE statement  to find out information about 

– file existence

– file unit open status

– various file attributes

The syntax has two forms, one based on file name, the 
other for unit number

INQUIRE(file='name', options ...)

INQUIRE(unit=iu, options ...)

File opening: file properties

EXIST does file exist ? (LOGICAL)

OPENED is file / unit opened ? (LOGICAL)

FORM 'formatted' or 'unformatted' (CHAR)

ACCESS 'sequential' or 'direct' or 'stream' (CHAR) 

ACTION 'read', 'write', 'readwrite' (CHAR)

RECL record length (INTEGER)

SIZE file size in bytes (INTEGER)

File opening: file properties example

! Find about the existence of a file 

LOGICAL :: file_exist

INQUIRE (FILE='foo.dat', EXIST=file_exist)

IF (.NOT. file_exist) THEN
WRITE(*,*) 'The file does not exist‘

ELSE
! Do something with the file 'foo.dat‘

ENDIF

File writing and reading

Writing to and reading from a file is done by giving the 
corresponding unit number (iu) as a parameter :
WRITE(iu,*) str

WRITE(unit=iu, fmt=*) str
READ(iu,*) str
READ(unit=iu, fmt=*) str

Formats and other options can be used as needed

If keyword 'unit' is used, also the keyword 'fmt' must be 
used

– Note: 'fmt' is applicable to formatted, text files only

The star format (*) indicates list-
directed output (i.e. programmer does 
not choose the input/output styles)

Formatted vs. unformatted files

Text or formatted files are

– Human readable

– Portable i.e. machine independent

Binary or unformatted files are

– Machine readable only, generally not portable

– Much faster to access than formatted files

– Suitable for large amount of data due to reduced file sizes

– Internal data representation used for numbers, thus no number 
conversion, no rounding of errors compared to formatted data

Unformatted I/O

Write to a sequential binary file
REAL rval

CHARACTER(len = 60) string
OPEN(10, file='foo.dat', form='unformatted')
WRITE(10) rval
WRITE(10) string
CLOSE(10)

No FORMAT descriptors allowed

Reading similarly
READ(10) rval

READ(10) string

Stream I/O

A binary file write adds extra record delimiters (hidden 
from programmer) to the beginning and end of records 

In Fortran 2003 using access method 'stream' avoids this 
and implements a C-like approach

– It is recommended to use stream I/O

Create a stream (binary) file
REAL dbheader(20), dbdata(300)

OPEN(10,file='my_database.dat', access='stream')
WRITE(10) dbheader

WRITE(10) dbdata
CLOSE(10)

Reading similarly

Summary

Input/Output formatting

Files: communication between a program and the 
outside world

– Opening and closing a file

– Data reading & writing

Use unformatted (binary) I/O for all except text files

Stream I/O

Internal I/O

Derived data types

112

Outline

Recalling Fortran built-in data types

Rationale behind derived data types

Data type declaration and visibility with examples

Fortran built-in types

Standard Fortran already supports a wide variety of 
fundamental data types to represent integers, floating 
point numbers ("real"), truth values ("logical") and 
variable length character strings

In addition each of these built-in types may have 
declared as multi-dimensional array

Furthermore, reals and integers can be declared to 
consume less memory in expense of reduced numerical 
precision through kind parameter (e.g. 8 or 4)

The variable representation method (precision) may be 
declared using the KIND statement

! SELECTED_INT_KIND(r)
! SELECTED_REAL_KIND(p)
! SELECTED_REAL_KIND(p,r)

INTEGER, PARAMETER :: short=SELECTED_INT_KIND(4)
INTEGER, PARAMETER :: double=SELECTED_REAL_KIND(12,100)
INTEGER (KIND=short) :: index
REAL (KIND=double) :: x, y, z
COMPLEX (KIND=double) :: c

x=1.0_double; y=2.0_double * ACOS(x)

Integer between -10r < n < 10r

Real number accurate up to p decimals

A floating point (real) number between
-10100 < x < 10100, accurate up to 12
decimals

A few words about numerical precision

Numerical precision: example

PROGRAM Precision_Test
IMPLICIT NONE

INTEGER, PARAMETER :: sp = SELECTED_REAL_KIND(6,30), &
dp = SELECTED_REAL_KIND(10,200)

REAL(KIND=sp) :: a
REAL(KIND=dp) :: b
WRITE(*,*) sp, dp, KIND(1.0), KIND(1.0_dp)
WRITE(*,*) KIND(a), HUGE(a), TINY(a), RANGE(a), PRECISION(a)
WRITE(*,*) KIND(b), HUGE(b), TINY(b), RANGE(b), PRECISION(b) 

END PROGRAM Precision_Test

Program output:

4  8  4  8

4 3.4028235E+38 1.1754944E-38 37 6 

8 1.797693134862316E+308 2.225073858507201E-308 307 15

Fortran 2008 ISO_FORTRAN_ENV

MODULE prec

USE ISO_FORTRAN_ENV, ONLY: INT32, INT64, REAL32, REAL64

IMPLICIT NONE

PRIVATE

INTERGER, PARAMETER :: i4 = INT32 &

i8 = INT64 &

r4 = REAL32 &

r8 = REAL64

PUBLIC :: i4, i8, r4, r8

END MODULE prec

Other intrinsic functions related to numerical precision

KIND(A)             Returns the kind of the supplied argument
TINY(A)             The smallest positive number
HUGE(A)           The largest positive number
EPSILON(A)         The smallest positive number added to 1.0 

returns a number just greater than 1.0
PRECISION(A)       Decimal precision
DIGITS(A)          Number of significant digits
RANGE(A)           Decimal exponent
MAXEXPONENT(A) Largest exponent (of the kind(A))
MINEXPONENT(A)  Smallest exponent (of the kind(A))

Numerical precision

What is a derived data type ?

Derived data type is a data structure composed of built-in 
data types and possibly other derived data types

– Equivalent to struct in C programming language

Derived type is defined in the variable declaration section 
of the programming unit

– By default not visible to other programming units

 Unless defined in a module and used via USE clause, which is 
most often the preferred way

Derived data types – rationale

Properly constructed data types make the program more 
readable, lead to clean interfaces and reduce errors

Variables used in the same context should be grouped 
together, using modules and derived data types

Please do not forget computationally efficient data layout 
when diving into object oriented programming in Fortran 
(or any other language)

– Structure of Arrays (SoA) versus Arrays of Structures (AoS)

Data type declaration: example

! Define a type describing a player

TYPE playertype
CHARACTER (LEN=30) :: name
INTEGER :: number, goals, assists

END TYPE playertype

! Declare variables of the player type

TYPE(playertype) :: ville, pekka

TYPE(playertype), DIMENSION(30) :: players

Data type access: example

! Initialization of / access to type components
ville % name = 'Ville Nieminen'
ville % number = 17
ville % goals = 10
ville % assists = 8

! Initialization via type constructor
ville = playertype('Ville Nieminen', 17, 10, 8)

! Arrays of derived types accessed elementwise
players(1) % name = 'Pekka Saravo'
players(1) % number = 6
players(1) % goals = 2
players(1) % assists = 4

Nested derived types: example

! Type declaration

TYPE hockeyteam
CHARACTER (LEN=80) :: name
TYPE(playertype) :: players(30)
TYPE(goalietype) :: goalies(3)

END TYPE hockeyteam

! Variable declaration
TYPE(hockeyteam) :: tappara, ilves, karpat

! Initialization / access
tappara % name = 'Tappara'
tappara % players(2) % name = 'Ville Nieminen'
tappara % players(2) % number = 17

Visibility of derived data types

When declared in the same programming unit derived 
data types are visible to that unit only

– and sub-units under CONTAINS statement

When declared in a module unit, a derived data type can 
be accessed outside the module through USE-statement

Summary

Derived data types enables grouping of data to form 
logical objects

A Fortran program becomes more readable and modular 
with sensible use of derived data types

Handling of linked lists or binary trees becomes more 
manageable with use of data structures

Enables the use of object oriented programming 
concepts

Useful new features

Outline

Fortran 2003 is a major extension over Fortran 90/95

– Fortran 2008 is a smaller extension over Fortran 2003, but 
still significant due to addition of co-arrays

F03 and F08 standardize features which were language 
extensions in many compilers

– reading command line arguments, getting values for 
environment variables, system calls, etc.

F03 brings improvements to semantics of ALLOCATABLE
and POINTER variables

Several small miscellaneous enhancements 

Command line arguments

Parameters to a program are very often given to programs as 
command line arguments
– Input file(s), modify program behavior, etc.

Fortran 2003 has (finally!) a standardized method for reading 
command line arguments
– No need to use extensions such as GETARG and IARGC

The standard function calls are, respectively
– get_command_argument and command_argument_count

In addition,  in order to access the whole command line, use

– get_command

128

Command line arguments …

Access separate command line arguments
call get_command_argument(number[,value][,length][,status])

– number is of type integer and denotes which argument to get

– value is of type character string and contains the value of the requested 
argument on return. If the actual argument is too short or long it is padded 
with blanks or truncated, respectively (optional)

– length is of type integer and contains the length of the requested argument 
on return (optional)

– status is of type integer. On successful return status is 0, -1 if value was 
too short to contain actual argument and 1 if argument could not be returned 
(optional)

Get the number of command line arguments
integer :: command_argument_count()

Command line arguments …

Access the whole command line
call get_command(command[,length][,status])
– command is of type character string and contains the value of the 

command line on return. If the the actual command line is too short or 
long it is padded with blanks or truncated

– length is of type integer and contains the length of the command 
line on return or zero if the length cannot be determined (optional)

– status is of type integer. On successful return status is 0, -1 if value 
was too short to contain actual argument and 1 if argument could not 
be returned (optional)

Command line arguments: example

program commandline
implicit none
character(len=256) :: line
integer :: i, iarg, stat, clen

! Individual arguments
iarg = command_argument_count()
do i=1,iarg

call get_command_argument(i,line,clen,stat)
write (*,'(I0,A,A)') i,': ',line(1:clen)

end do

! The whole command line
call get_command(line,clen,stat)
write (*,'(A)') line(1:clen)

end program commandline

Environment variables

Besides command line arguments, environment variables 
are a common way to modify program behaviour

Fortran 2003 has (finally!) a standardized method for 
accessing values of environment variables

In Fortran 77/90/95 accessing getenv from C standard 
library requires passing character strings from Fortran to 
C and back

– Without ISO-C bindings (only in Fortran 2003/2008), error 
prone and nonportable methods needed

Environment variables

Access a value of an environment variable
call get_environment_variable(name,value[,length]

[,status][,trim_name])
– name is of type character string and contains the name of the requested variable

– value is of type character string and contains the value of the requested variable. If the 
the actual variable value is too short or long it is padded with blanks or truncated. If the 
variable has no value or does not exist, value is set to blanks

– length is of type integer and contains the length of the requested variable on return if 
the variable exists and has a value and zero otherwise (optional)

– status is type integer. If requested variable does not exist status is 1. If value was too 
short status is -1 and zero otherwise. For other return codes, see docs (optional)

– trim_name is of type logical and sets if trailing blanks are allowed in variable names or 
not (optional)

Environment variables: example

program environment
implicit none
character(len=256) :: enval
integer :: len,stat

! Extract HOSTNAME
call get_environment_variable('HOSTNAME',enval,len,stat)
if (stat == 0) write (*,'(A,A)') 'Host=', enval(1:len)

! Extract USER
call get_environment_variable('USER',enval,len,stat)
if (stat == 0) write (*,'(A,A)') 'User=', enval(1:len)

end program environment

Executing commands

Invoking external programs from within a program is 
occasionally needed

– No source nor library API available for a useful program

Fortran 2008 has (finally!) a standardized method for 
invoking an external command

In Fortran 77/90/95 accessing system from C standard 
library requires passing character strings from Fortran to 
C and back

– With Fortran 2003, ISO-C bindings can be used

Executing commands

Execute a command line
call execute_command_line(command[,wait][,exitstat]

[,cmdstat][,cmdmsg])

– command is a character string containing the command to be invoked

– wait is logical value indicating if command termination is to be waited 
(.true., the default) or if the command is to be executed 
asynchronously (.false.) (optional)

– exitstat is an integer value containing the return value of the 
command if wait=.true. (optional)

– cmdstat is an integer value. It is assigned a value of zero if command
executed successfully. For other return codes, see docs (optional)

– cmdmsg is a character string containing explanatory message for 
positive values of cmdstat (optional)

Executing commands: example

program execcommand
implicit none
integer :: estat, cstat

! Execute unix-command 'ls -al'
call execute_command_line('ls -al', .TRUE., estat, cstat)
if (estat==0) write (*,'(A)') 'Command completed successfully’

end program execcommand

Enhancements to dynamic memory allocation 

Fortran 2003 added features related to dynamic memory 
allocation

The most common are

– Automatic DEALLOCATE when ALLOCATABLE goes out of 
scope

– ALLOCATABLE components in derived data types

– ALLOCATABLE procedure arguments

– ALLOCATABLE function return values

A variable declared as an 
ALLOCATABLE local 
variable, is automatically 
deallocated upon 
returning back from 
routine (goes “out of 
scope”)

The rule holds for 
variables without the 
SAVE –attribute

SUBROUTINE SUB1
INTEGER, ALLOCATABLE :: A(:)
ALLOCATE(A(1000))
CALL USE_ARRAY(A)
! DEALLOCATE(A) ! Not needed

END SUBROUTINE SUB1

SUBROUTINE SUB2
INTEGER, ALLOCATABLE, SAVE :: A(:)
ALLOCATE(A(1000))
CALL USE_ARRAY(A)
DEALLOCATE(A) ! Is still needed

END SUBROUTINE SUB2

Automatic DEALLOCATE

In Fortran90 dynamic 
components of user 
defined data types were 
restricted to have the 
POINTER attribute

The restriction was 
removed in Fortran 2003

MODULE my_mod
TYPE my_type
REAL, ALLOCATABLE :: ARRAY(:)

END TYPE my_type
END MODULE my_mod

SUBROUTINE SUB(N)
USE my_mod
IMPLICIT NONE
INTEGER, intent(in) :: N
TYPE(my_type) :: T
:
ALLOCATE(T % ARRAY(N))
:

END SUBROUTINE SUB

ALLOCATABLE type components

Procedure arguments are 
permitted to be 
ALLOCATABLE

Dynamic memory 
allocation in the callee
based on sizing 
information

Allocated (and possibly 
initialized) data returned 
to the caller

PROGRAM TEST
INTEGER :: N
REAL, ALLOCATABLE :: A(:)
CALL SUB(A, N)
CALL USE_ARRAY(A, N)
DEALLOCATE(A)

END PROGRAM TEST

SUBROUTINE SUB(A, N)
REAL,ALLOCATABLE,INTENT(OUT)::A(:)
INTEGER, INTENT(OUT) :: N
READ *, N
ALLOCATE(A(N))
READ *, A
! No automatic DEALLOCATE here!

END SUBROUTINE SUB

ALLOCATABLE procedure arguments

Functions are permitted 
to have ALLOCATABLE
value

Functionality similar as 
with procedure arguments

Particularly useful when a 
dynamically sized array is 
returned

PROGRAM TEST
REAL :: UNPACKED(1000)
REAL, ALLOCATABLE :: A(:)
A = PACK_DATA(UNPACKED)
CALL USE_PACKED_DATA(A)
DEALLOCATE(A)

CONTAINS
FUNCTION PACK_DATA(X) RESULT (PK)
REAL, ALLOCATABLE :: PK(:)
REAL, INTENT(IN) :: X(:)
INTEGER :: NPK
NPK = NUM_DISTINCT_VALUES(X)
ALLOCATE(PK(NPK))
CALL GET_DIST_VALS(X,...,PK,NPK)

END FUNCTION PACK_DATA
END PROGRAM TEST

Function return value as ALLOCATABLE

Typically used with 
CHARACTER strings, 
whose length is decided at 
run time

Allocation can either 
explicit or automatic

– Automatic allocations 
may require additional 
compiler flags

PROGRAM TEST
CHARACTER(:), ALLOCATABLE :: CH
INTEGER :: LENCH
LENCH = LEN('Hello World!')
! Explicit allocation
ALLOCATE(character(LENCH)::CH)
CH(:) = 'Hello World!‘
! Output: 12
PRINT *,LEN(CH)
! Legal, but not required
DEALLOCATE(CH)

END PROGRAM TEST

ALLOCATABLE scalars

Moves allocation from 
one memory location to 
another

Previous allocation 
becomes unallocated and 
new allocation holds the 
previous data

Arrays to (=A) and from 
(=B) must have the same 
type and rank

PROGRAM TEST
INTEGER, ALLOCATABLE :: A(:), B(:)
ALLOCATE(A(1:5)) ; A(:) = 0
A(3) = 3
! Output: 0 0 3 0 0
PRINT *, 'A=',A
CALL MOVE_ALLOC(A,B)
! A is now DEALLOCATED
! B is now ALLOCATED
IF (ALLOCATED(A)) PRINT *, '>A=',A
IF (ALLOCATED(B)) PRINT *, '>B=',B
! Legal, but not required
DEALLOCATE(B)
! Would be illegal: DEALLOCATE(A)

END PROGRAM TEST

Transferring an allocation: MOVE_ALLOC

Summary

With Fortran 2003 and 2008, it is possible to obtain command 
line arguments, access environment variables and run 
operating system commands in a standardized way

Dynamic memory allocation has been improved by means of 
enhancements to ALLOCATABLE variables

New edit descriptors enable dynamic sizing of output

Language interoperability

Outline

Language interoperability issues

Module ISO_C_BINDING

– Mapping of C intrinsic data types in Fortran

– Mapping of C derived data types in Fortran

– Utility functions

Calling C routines from Fortran

Mapping of array data

Mapping of character data

Accessing global data from Fortran

147

Language interoperability issues

The problem of language interoperability has been 
present for a long time

Consider the following scenario:

– An application program is written entirely in Fortran

– A library, written entirely in C by some third party, would 
be useful in the Fortran program

– How to exploit the library written in C in a portable way?

Traditional interoperability, i.e., “do not use this”

By using the facts that Fortran compilers...

– Pass all procedure arguments by-reference (i.e. by-address)

– Usually refer to functions and subroutines in lowercase letters 
and by adding an additional underscore after, e.g. Func
becomes func_ to be found by the linker

– Usually pass function return values via stack 

– Usually pass CHARACTER-strings by-reference, with an 
additional hidden length argument, passed by value. NOTE: 
Fortran CHARACTER-strings are not null-character terminated

…C code can be referenced from Fortran

149



Traditional interoperability

Consider a case, where CBLAS-function is to be called in 
order to calculate a dot product in single precision

The C function prototype for “cblas_sdot“ from 
<cblas.h> is:
float cblas_sdot( const int N,

const float *X, const int incX,
const float *Y, const int incY)

In order to call this from Fortran – using traditional, not-
recommended approach – an extra wrapper layer is 
needed

Traditional interoperability: example

! F_calls_C.F90

program F_calls_C
implicit none
integer, parameter :: N=100
real :: X(N), Y(N), SDOT
REAL, external :: SDOT_FUNC

! Init X & Y
X(:) = 1.5e0; Y(:) = 2.5e0
! Wrapper call
SDOT = SDOT_FUNC(X, Y, N)
write (*,*) 'SDOT=', SDOT

end program F_calls_C

/* cblas_sdot_wrap.c */

/* This approach works in 99% of the 
cases in Unix. */ 

#include <cblas.h>

float sdot_func_(const float *X,
const float *Y,
const int *N)

{ /* Wrapper-code */
/* Actual call to CBLAS-routine */
float sdot = cblas_sdot(*N,

X, 1,
Y, 1);

return sdot;

}

Interoperability with Fortran 2003

The traditional way to have interoperability with C is a 
non-portable hack, i.e., it requires a-priori knowledge of 
lowercase and underscore policy used by the compiler

Complex cases, such as passing CHARACTER–strings or 
passing arguments by value, are generally very error 
prone and may lead to catastrophic errors at runtime

With Fortran 2003, interoperability with C is defined in 
the standard by using INTERFACE-blocks with the 
bind(C)-attribute

Interoperability with Fortran 2003: example

program F_calls_C
use, intrinsic :: ISO_C_BINDING
implicit none
integer(kind=C_INT), PARAMETER :: N = 100
real(kind=C_FLOAT) :: X(N), Y(N), SDOT
interface
function cblas_example(N, X, INCX, Y, INCY) &

bind(c,name='cblas_sdot') RESULT(anumber)
use, intrinsic :: ISO_C_BINDING
integer(kind=C_INT), value :: N, INCX, INCY  ! Pass by value
real(kind=C_FLOAT), intent(in) :: X(*), Y(*) ! intent(in)=const float *
real(kind=C_FLOAT) :: anumber

end function cblas_example
end interface
! Init X & Y
X(:) = 1.5e0; Y(:) = 2.5e0
SDOT = cblas_example(N, X, 1, Y, 1) ! A direct call to ’cblas_sdot’
write (*,*) 'SDOT=',SDOT

end program F_calls_C

The ISO_C_BINDING module

Fortran 2003  intrinsic module ISO_C_BINDING is used –
as we have already seen – with
USE, INTRINSIC :: ISO_C_BINDING

Module contains

– Access to named constants that represent kind type 
parameters of data representations compatible with C-types. 

– The derived types C_PTR and C_FUNPTR corresponding to C 
pointer and C function pointer types, respectively

– Useful procedures: C_LOC, C_FUNLOC, C_F_POINTER, 
C_ASSOCIATED, C_F_POINTER, C_F_FUNPOINTER, 
C_SIZEOF (F08)

Calling C routines 

A Fortran SUBROUTINE maps to a C-function with void
result

A Fortran FUNCTION maps to a C-function returning a 
value

Binding label in bind(c, name=<label>)

– The routine is known to the C compiler as specified by the 
binding label

– By default the Fortran name in lower case. If provided, 
case sensitive ignoring leading and trailing blanks 

(name='C_funcX')

Mapping of C intrinsic data types

Interoperable mappings for the most commonly used C 
intrinsic data types

Traditional “old” Fortran Fortran declaration C data type

INTEGER*2 INTEGER(c_short) short int

INTEGER*4 INTEGER(c_int) int

INTEGER*8 INTEGER(c_long_long) long long int

REAL*4 REAL(c_float) float

REAL*8 REAL(c_double) double

CHARACTER*1 CHARACTER(1,c_char) char

Mapping of derived data types

In many cases it is possible to describe Fortran derived 
data types in terms of C data structures (and vice versa)

To be interoperable, Fortran derived type must have the 
bind(c) attribute

– sequence or extends keywords are forbidded

Individual Fortran components in the data type must be 
of an interoperable type

– zero-sized array components are forbidden

– ALLOCATABLE and POINTER components are forbidden

C types cannot be unions nor structures with bit-fields 

Mapping of derived data types

For a derived type to be interoperable between C and 
Fortran, variable ordering, data types and array sizes 
must be identical

– Variable names do not need to be identical

Typical usage comes through function calls, for instace a 
Fortran routine extracting information from a C routine

Mapping of derived data types: example

/* C data structure */

typedef struct {
int count;
double d[100]; 

} C_type;

/* C-function example */

void C_func(C_type *p) {
p->count = 1;
p->d[0] = 1.23;

}

program typedeftest
use, intrinsic :: ISO_C_BINDING
implicit none
type, bind(c) :: C_type_in_Fortran
integer(kind=c_int) :: count
real(kind=c_double) :: d(100)

end type C_type_in_Fortran
interface
subroutine testf(P) &

bind(c, name='C_func')
import
type(C_type_in_Fortran) :: P

end subroutine testf
end interface
type(C_type_in_Fortran) :: X
call testf(X)
write (*,*) X % count, X % d(1)

end program typedeftest

Array indexing

– C: starts from zero (0)

– Fortran: by default, starts 
from one (1)

Multidimensional ordering

– C: “row-major”, grow 
fastest along the last 
dimension 

– Fortran: “column-major”,  
grow fastest along the 
first dimension

! Fortran array declarations
! compatible with C
REAL (c_double) :: z1(5), z2(3:5,17)
INTEGER (c_int) :: ivec(-4:7)

Mapping of array data

/* Corresponding C-declarations */

double z1[5], z2[17][3];
int  ivec[12];

Mapping of character data

C procedures expect character strings to be null 
terminated

– A null character has to be appended to a string before 
calling the C procedure

Module ISO_C_BINDING contains many character 
constants: C_ALERT, C_BACKSPACE, C_FORM_FEED, 
C_CARRIAGE_RETURN, C_HORIZONTAL_TAB, 
C_NULL_CHAR, C_NEW_LINE, C_VERTICAL_TAB

Mapping of character data: example

program F_atoi
use, intrinsic :: ISO_C_BINDING
implicit none
interface
! C-prototype: int atoi(const char *num);
function atoi(num) bind(c, name='atoi') RESULT(x)

use, intrinsic :: ISO_C_BINDING
character(kind=c_char), INTENT(IN) :: num(*)
integer(kind=c_int) :: x

end function atoi
end interface
character*(*), parameter :: year = '2013'
character(:,kind=c_char), allocatable :: number
allocate(character(len=len(YEAR)+1)::number) ! Space for c_null_char
number = trim(year) // c_null_char
write (*,'(A,I0)') 'atoi('//year//')=',atoi(number)

end program F_atoi

Accessing global data

Global data:

– Fortran: is declared in modules or in COMMON blocks

– C: is declared outside any local scope and referenced via 
extern elsewhere

C global data is accessed in Fortran with
bind(c, name=<label>)

– Optional name -parameter is defined similarly as when 
accessing C procedures

– NOTE: using bind(c) implies save

Accessing global data: example

/* Global C data */

int number; 

float Array[8];

double slice[3][7];

struct coord {
float x, y, z;

};

struct coord xyz; 

! Fortran global data must be
! defined in a module

module something
use, intrinsic :: ISO_C_BINDING
implicit none

integer(c_int), bind(c) :: number
real(c_float) :: my_array(8)
bind(c, name=‘Array’) :: my_array

! Index swap
real(c_double), bind(c) :: & 

slice(7,3)
real(c_float) :: x, y, z
common /xyz/ x, y, z
! Note /…/ syntax
bind(c) :: /xyz/

end module something

Summary

With ISO_C_BINDING, Fortran has standardized 
mechanisms to access source code and libraries written 
in C, as well as define Fortran accessible from C

Interoperability should be used with well-defined 
interfaces

– Complicated structures or calling sequences are not 
recommended

Outlook on Fortran 2003/2008

166

Outline

Fortran 2003 and Fortran 2008 in short

Abstract interfaces and procedure pointers 

Object-oriented features in Fortran 2003/2008

Fortran coarrays (CAF)

Fortran 2003 in short

Fortran 2003 is a major revision over Fortran 95 standard
– Interaction with the environment

– Interoperability with C

– I/O enhancements

– Object-oriented features

– Support for IEEE floating point arithmetics

– Several minor enhancements

Fortran 2008 in short

Fortran 2008 upgrades the standard even further

– Fortran coarray (CAF) syntax

– Several new intrinsic procedures (bit manipulation etc.)

– Several performance related enhancements

– Several minor enhancements and clarifications

In Fortran 90/95, 
interfaces for different 
procedures must be 
declared separately, even 
if the procedures have the 
same external interface

Fortran 2003 allows 
declaration of abstract
interfaces

Referenced with the 
procedure statement

! Abstract interface definition
abstract interface

function trig(x) result(y)
real, intent(in) :: x
real :: y

end function trig
end interface

! Procedure definition 
procedure(trig) :: mysin
write (*,*) mysin(x)

Abstract interfaces

The procedure -
statement allows also the 
definition of pointers to 
procedures (explicit or 
implicit interface)

Procedure pointers offer 
similar functionality as 
function pointers in the C 
programming language

! Interface definition
abstract interface

function trig(x) result(y)
real, intent(in) :: x
real :: y

end function trig
end interface

procedure(trig), pointer :: p1
p1 => mysin
write (*,*) p1(x) ! Prints mysin(x)

Procedure pointers

Object-oriented programming in Fortran

Fortran 2003/2008 supports object-oriented (OO) 
programming

– Abstract interfaces

– Type extensions, polymorphism (single inheritance), 
abstract types

– Type and- object-bound procedures, type operators, type-
bound generics, finalizers

Object model designed to maintain backwards 
compatibility with Fortran 95

Type extensions are used 
to extend the functionality 
of an existing type

Type extensions are 
backwards compatible, 
because every instance of 
extended is also an 
instance of base

type :: base
integer :: field1

end type base

type, extends(base) :: extended
integer :: field2

end type extended

Type extension

Polymorphism = ”The 
provision of a single 
interface to entities of 
different types”, Bjarne 
Stroustrup, creator of C++

Object polymorphism 
(defined by the class
keyword): Different 
instances from different 
classes are related by a 
common base class

type :: base
integer :: field1

end type base

type, extends(base) :: extended
integer :: field2

end type extended

type(extended), target :: t1
class(base), pointer :: p1

t1 = extended(1,2)
! p1 is of class base
p1 => t1

Polymorphism

Procedures as type components

In the object-oriented programming paradigm, 
procedures acting on the data are tied to the data

Invocation of procedures for polymorphic variables via a 
runtime decision (dynamic dispatch)

In Fortran 2003 procedures can be tied to

– objects (object-bound), i.e., procedure pointer 
components

– types (type-bound), i.e., procedure components, can be 
overriden

Type-bound procedures 
are dynamically deter-
mined for polymorphic 
variables during runtime 

Determination of which 
type-bound procedure to 
call has a smaller 
performance penalty than 
a full type determination

type :: base
integer :: field1
contains
procedure :: &
write_output => write_base

end type base

type, extends(base) :: extended
integer :: field2
contains
procedure :: &
write_output => write_extended

end type extended

! Implementations of write_base and
! write_extended omitted

Type-bound procedures

In Fortran 2003 types can 
contain generic interfaces 
to procedures and 
operators

Enables ”natural” use of 
user determined types in 
the program

module mymodule
type myreal
real :: value
contains
procedure :: add => myadd
generic :: operator(+) => add
end type
! Other interfaces to operations
contains
function myadd(x,y) result(z)
implicit none
class(myreal), intent(in) :: x, y
type(myreal) :: z
z % value = x % value + y % value
end function myadd 
! Other operations with myreal

end module mymodule

GENERIC type-bound procedures

An abstract type creates a 
base type for all the 
extending types to build 
upon

Can be also used to create 
common interfaces for all 
the extending classes to 
implement

type, abstract :: abs_base
end type abs_base

type, extends(abs_base) :: extended
integer :: field1

end type extended

type(extended), target :: t1
class(abs_base), pointer :: p1

t1 = extended(i)
p1 => t1

Abstract types

final procedures of a 
derived type are 
automatically called 
whenever the type is 
deallocated

Typically used for 
housekeeping duties 
(closing file handles, 
deallocating pointer 
components etc.)

type handle
private
! Data containing handle
contains
final :: close_handle

end type handle

subroutine close_handle(this)
implicit none
class(handle) :: this
! Implementation omitted

end subroutine close_handle

FINAL components

Fortran coarrays (CAF)

Parallel processing as part of Fortran language standard

– Only small changes required to convert existing Fortran code 
to support a robust and potentially efficient parallelism

A Partitioned Global Address Space (PGAS) paradigm

– Parallelism implemented over “distributed shared memory”
 CAF is potentially massively parallel

Integrated into Fortran 2008 standard

– Compiler support is still incomplete (Cray: excellent, Intel: 
moderate, GNU: barely compiles)

CAF in short

Adds only a few syntactic alterations and rules to the 
Fortran language

Provides mechanisms to allow

– SPMD (Single Program, Multiple Data) style of explicitly 
parallel programming

– Data distribution over partitioned memory

– Synchronization primitives

– Memory management for dynamic shared entities

Execution model

Upon startup a Co-Array Fortran program gets replicated 
into a number of copies called images (i.e. processes)

– The number of images is usually decided at the execution 
time, but (in rare cases) can also be fixed at compile time

Each “replica” (image) runs asynchronously in a loosely 
coupled way until program controlled synchronization

Image’s (local) data are visible within the image only –
except for data declared as special arrays i.e. co-arrays

– One-sided data communication enables movement of co-
array data across different images of a CAF program

Program replicated into 
separated copies called 
images

Each image executes 
concurrently, 
synchronization with 
separate calls

! Hello world with CAF
program hello

implicit none
write (*,*) 'Hello from ', &

this_image(),  &
' out of',     &
num_images()

end program hello

Execution model

Division of data to 

– global: visible to all 
images, ”[]” operator

– local: visible only locally, 
”()” operator

Global data is read/write 
accessible to all images

program data
integer :: global(3)[*], &

local(3), npes
npes = num_images()
! Initialize global data
global(:) = this_image()*(/1,2,3/)
sync all
! Initialize local data by copying
! from image number “npes”
local(:) = global(:)[npes]

end program data

Data distribution

Data distribution: example

Assume 4 images with 3 elements on each image

Local array local=[4,8,12] on each image

Co-array global is

1 2 3 2 4 6 3 6 9 4 8 12global =

1 2 3 2 4 6 3 6 9 4 8 12

Image#1 Image#2 Image#3 Image#4

global(:)[1] global(:)[2] global(:)[3] global(:)[4]

Accesses to global data 
need to be carefully 
considered in order to 
avoid race conditions

Synchronization 
statements 

– global:
sync all

– subset of images: 
sync images(list)

! First image synchronizes with all,
! the rest with the first
if ( this_image() == 1 ) then

sync images(*) ! “*” = all
else

sync images(1)
end if

! Do operations on global data
…
! Synchronize globally
sync all

Synchronization

critical construct 
ensures that a block of 
code is executed only by 
one image at a time

In order to ensure that all 
the images have done 
their work within a critical 
section, explicit 
synchronization is needed

program crittest
implicit none
integer :: mysum[*] = 0
integer :: me, npes

me = this_image()
npes = num_images()
! Compute global sum to image 1
critical
mysum[1]=mysum[1]+me

end critical
! Wait for all images to finish
sync all
if (me==1) write (*,*) mysum

end program crittest

CRITICAL sections

Compared to critical
sections, lock variables 
allow more fine grained 
access control

Acquired with lock, 
release with unlock

lock_type is defined in 
the intrinsic module 
iso_fortran_env

program locktest
use, intrinsic :: iso_fortran_env
implicit none
type(lock_type) :: lockvar[*]
integer :: mysum[*] = 0, me, npes
me = this_image()
npes = num_images()
! Acquire lock
lock(lockvar[1])
! Critical access to mysum[1]
mysum[1]=mysum[1]+me
! Release lock
unlock(lockvar[1])
! Wait for all images to finish
sync all
if (me==1) write (*,*) mysum

end program locktest

LOCK variables

Summary

Fortran is a modern language that continues to evolve

Fortran remains as one of the major languages for 
scientific computation

A more thorough introduction to Fortran object-oriented 
features and coarrays are given during the course 
”Advanced Fortran Topics & Coarray Fortran”

